

## Un amplificateur avec sortie Push Pull CMOS

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDE](#)

Le schéma analysé associe bipolaire et CMOS pour réaliser un amplificateur de puissance. Les composants sont issus de Eval.lib. Les performances de cet ampli audio restent modestes : à 1 kHz, rendement >50 %, distorsion < 5 %.

### 1) Description de l'amplificateur

Il est alimenté en mono polaire par  $E = 40 \text{ V}$ .

R1 = 1 kΩ  
R2 = 27 kΩ  
R3 = 100 Ω  
R4 = 470 Ω  
R5 = 2,4 kΩ  
R6 = 1,8 kΩ  
R7 = 330 Ω  
R8 = 330 Ω  
RL = 8 Ω

C1 = 220 μF  
C2 = 1000 μF  
C3 = 2200 μF

Q1 :  
Q2N2222  
M1 :  
IRF150  
M2 :  
IRF9140  
E = 40 V




Schéma analysé

- On trouve classiquement des condensateurs de liaison (C1, C3) qui coupent la composante continue afin que  $V_{in}$  ou  $R_L$  ne perturbe pas la polarisation.
- Q1 est monté en **émetteur commun**, par C2 qui court-circuite R3 en dynamique. Cet étage apporte l'amplification en tension. La polarisation est telle que Q1 est en classe A.
- L'étage de puissance est assuré par le **push pull** M1, M2, transistors MOS, qui réalise une amplification en courant, pour pouvoir débiter sur faible impédance (charge  $R_L = 8 \Omega$ ).

L'amplificateur résultant est un « ampli audio » dont nous allons chiffrer les performances.

### 2) Analyse de la polarisation

#### 2.a) Prédétermination

- On remarque que Q1 est autopolarisé : c'est la tension de sortie (point milieu  $\approx 20 \text{ V}$ ) qui polarise la Base par la branche R2.

Avec l'hypothèse que  $\beta R3 \gg R6$ , (ou  $I_B$  négligeable devant  $I_{R6}$ ), le potentiel de base sera :  $V(6) R6 / (R2+R6)$ , soit :  $20 \times 1,8 / (27+1,8) = 1,25 \text{ V}$ . En prenant  $V_{BE} = 0,6 \text{ V}$ , on déduit  $U_{R3} = 0,65 \text{ V}$ , soit un courant de collecteur voisin de 6,5 mA.

La tension attendue aux bornes de R4 est  $R4 I_C = 0,47 \times 6,5 = 3,05 \text{ V}$ . Elle fournit les 2 tensions  $V_{GS}$ . Celle aux bornes de R5 est  $R5 I_C = 2,4 \times 6,5 = 15,6 \text{ V}$ .

Le potentiel de collecteur est donc  $E - U(R4) - U(R5) = 40 - 3,05 - 15,6 = 21,35 \text{ V}$ .

Le potentiel d'émetteur est  $R3 I_C = 0,1 \times 6,5 = 0,65 \text{ V}$ . Il vient  $V_{CE} = 21,35 - 0,65 = 20,7 \text{ V}$ .

- Le push pull doit être polarisé en mode « source de courant contrôlée par une tension », ou, pour utiliser le vocabulaire du transistor à effet de champ, en mode « saturé ».

Pour ce faire, il faut à minima que  $UR5 > VT0_N + |VT0_P| = 2,83 + 3,67 = 6,5$  V de façon à les rendre passant. De plus, il faut  $V_{DS} > V_{GS} - V_{TH}$  pour amener à l'état saturé ( $V_{TH}$  est fonction de  $VT0$  et de la tension  $V_{DS}$ . Remarquons que  $V_{TH} \approx VT0$ ).

Or on a 3,05 V, ce qui rend impossible cette polarisation des 2 transistors, même en limite de conduction. Ce choix est justifié, car une polarisation en classe A consomme de la puissance. Remarquons que l'autopolarisation impose que M1 soit néanmoins conducteur. En conséquence c'est M2 qui sera bloqué. C'est le signal qui le mettra en conduction. Il travaille donc en classe B. La contrepartie est une distorsion de croisement, comme nous allons le voir par la simulation temporelle.

### 2.b) Simulation .OP

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
( 1 )	1.3233	( 2 )	1.3233	( 3 )	24.8210	( 4 )	21.8490
( 5 )	.6354	( 6 )	21.9680	( E )	40.0000	( G1 )	24.8210
( G2 )	21.8490	( IN )	0.0000	( OUT )	0.0000		

La sortie du push pull V(4) est à  $\approx 22$  V.

La résistance R2 est sous  $V(6) - V(2) = 21,968 - 1,3233 = 20,6447$  V.

Elle est parcourue par  $20,6447 / 27k = 0,764$  mA.

La résistance R6 est parcourue par  $1,3233 / 1,8 k = 735$   $\mu$ A.

On vérifie également que les transistors sont correctement polarisés :

NAME	Q1
MODEL	Q2N2222
IB	2.95E-05
IC	6.32E-03
VBE	6.88E-01
VBC	-2.05E+01
VCE	2.12E+01
BETADC	2.15E+02
GM	2.40E-01
RPI	9.52E+02
RX	1.00E+01
RO	1.50E+04
CBE	1.36E-10
CBC	2.33E-12
CJS	0.00E+00
BETAAC	2.29E+02

Bipolaire :

Courant de base = 29,5  $\mu$ A.

Courant de collecteur = 6,32 mA, valeur proche de celle déterminée avec beaucoup d'approximations.

$V_{BE}$  direct,  $V_{BC}$  inverse.

On a :

$g_m = 0,24$  (A/V).

RPI = 952  $\Omega$

R0 = 15 k $\Omega$

NAME	M1	M2
MODEL	IRF150	IRF9140
ID	8.14E-04	-4.94E-05
VGS	2.85E+00	-1.19E-01
VDS	1.80E+01	-2.20E+01
VBS	0.00E+00	0.00E+00
VTH	2.83E+00	-3.67E+00
VDSAT	2.24E-02	0.00E+00

ID(M1) = 814  $\mu$ A.

M1 est conducteur par la contre réaction R2.

ID(M2)  $\approx 50$   $\mu$ A (sortant) : on retrouve ID(M1) - I(R2)

M2 est quasi bloqué.

## 3) Amplification

### 3.a) Prédétermination de l'amplification petit signal de l'émetteur commun

Posons  $RC = (R4+R5) // R0$ . En milieu de bande, l'amplification de l'étage émetteur commun  $v(4)/v(2)$  est théoriquement :  $-g_m RC$  (L'influence de M1 et M2 est négligeable en milieu de bande).

Application numérique :

$$g_m = 0,24$$

$$RC = (0,6 k + 2,4 k) // 15 k = 2,5 k\Omega$$

L'amplification attendue est :

$$-0,24 \times 2500 = -600 \text{ soit } 55,5 \text{ dB.}$$


Posons  $REQ = RPI // R2 // R6 = 952 // 27\text{ k} // 1,8\text{ k} = 608\ \Omega$ .

La présence de R1 provoque un pont diviseur de tension, qui insère un coefficient  $REQ/(R1+REQ) = 608 / (1000 + 608) = 0,378$ , soit une perte de 8,44 dB entre  $v_{in}$  et la base de Q1.

La bande passante de cet émetteur commun est liée à l'impédance d'entrée du push pull. Cette dernière est liée aux condensateurs CGS, CGD de M1, M2, eux-mêmes liés à la polarisation.

On suppose que le push pull se comporte en suiveur parfait.

Le montage peut alors se modéliser par un amplificateur de coefficient A et d'impédance d'entrée  $Re (= RPI // R6 = 952 // 1,8\text{ k} = 622\ \Omega)$ , d'une résistance d'entrée R1, et d'une résistance de retour R2.




On peut exploiter le formalisme des schémas blocs. L'identification est :

$$a = -R'2 / (R1 + R'2) \text{ où } R'2 = R2 // Re = 27\text{ k} // 622 = 608\ \Omega$$

$$= -608 / (1000 + 608) = -0,378$$

$$b = R'1 / (R'1 + R2) \text{ où } R'1 = R1 // Re = 1\text{ k} // 622 = 383\ \Omega$$

$$= 383 / (383 + 27\text{ k}) = 0,014$$



On a, toujours en milieu de bande, la FTBF qui est alors classiquement :


$$v_{out} / v_{in} = a \frac{A}{1 + A b} = -0,378 \frac{600}{1 + 600 \times 0,014} = -24,12.$$

Cette prédétermination permet de voir que le gain est réglable par R2 (associée à R1).

### 3.b) Analyse harmonique de l'émetteur commun par la simulation

L'analyse AC de tout le circuit n'est pas réaliste par le fait que M2 n'est pas conducteur au point de polarisation.

On peut vérifier néanmoins la **réponse de l'étage Q1 seul**. Pour ce faire, il faut déplacer R2 : on retire la contre réaction, et on ajoute une tension continue de 21,968 V pour maintenir la polarisation à l'identique.




**Modification du schéma pour analyser l'étage Q1.**

La réponse harmonique résultante montre :

X Values	1.0000K
DB(V(4))	46.705
DB(V(4)/V(2))	55.108

Le gain est celui attendu : 55 dB pour l'émetteur commun seul.




**Réponse harmonique de l'émetteur commun**

Dans cette simulation AC de Q1, M1 et M2 ne sont pas polarisés dans les conditions nominales.

La fréquence de coupure haute n'est donc pas significative. La fréquence de coupure basse est 20 Hz

### 3.c) Simulation de l'amplificateur complet

Le signal d'entrée  $v_{in}$  est une sinusoïde à 1 kHz, de faible amplitude (pour rester dans le domaine « petits signaux »). Choisissons 50 mV, soit 0,1 V càc.




A cette échelle on distingue très bien le défaut de raccordement sur la tension de sortie, créé par M2 qui n'entre en conduction que par une valeur suffisamment élevée de  $v_{in}$  (ici, 20 mV).

On remarque que les transistors n'agissent pas de façon symétrique, car l'alternance positive a une plus grande amplitude que l'alternance négative.

## 4) Simulation temporelle, entrée à fort niveau

### 4.a) Amplification

On reprend l'essai précédent, mais avec une amplitude de 0,6 V, soit 1,2 V càc.



Le défaut de raccordement reste présent, mais est moins visible à cette échelle.

On mesure 26,6 V càc, soit, pour une entrée de 1,2 V càc, une amplification de 22,16 en opposition de phase : on est à 10 % de la prédétermination.

Pour information, en V(4), le signal est quasi sinusoïdal avec une dynamique de 3,67 V à 38,3 V. L'amplification apparente de l'émetteur commun est  $(38,3 - 3,67) / 1,2 = 28,86$ .

Remarque : un essai à 20 kHz, 0,6 V d'amplitude a donné un signal de sortie de 25,7 V càc, mais dégradé.

### 4.b) Bilan des puissances

Pspice propose l'affichage des puissances instantanées. Par calcul de valeur moyenne, on peut en déduire les puissances efficaces :

- 21,81 W délivrés par l'alim.
- 4,19 W dans M1,
- 5,87 W dans M2,
- 11,35 W dans RL.

X Values	10.000m
-AVGX(W(VDD),1m)	21.807
AVGX(W(M1),1m)	4.1912
AVGX(W(M2),1m)	5.8669
AVGX(W(RL),1m)	11.350

Rendement :


$$11,35 / 21,85 = 52 \%$$

### 4.c) Taux de distorsion

On conserve en  $v_{in}$  une entrée de 0,6 V d'amplitude, et de fréquence 1 kHz. Le signal de sortie n'est pas un sinus pur, et sa déformation s'exprime par le taux de distorsion.

Pour le quantifier, 2 solutions :

→ **par la FFT**, qui montre :  
 - la raie à la fréquence 1 kHz, **fondamentale**, appelée  $A_0$  dans la relation ci-dessous,  
 - toutes les raies constituant le signal.  
 Un signal sans distorsion ne serait composé que du fondamental à 1 kHz.



la FFT montre un fondamental et des harmoniques

Le taux de distorsion harmonique total (THD) est le rapport des harmoniques ajoutées (donc celles à 2 kHz, 3 kHz, 4 kHz..., appelées  $A_1, A_2, A_3, \dots$  dans la relation ci-dessous) et la valeur efficace du signal.

$$THD = \frac{\sqrt{A_1^2 + A_2^2 + A_3^2 + A_4^2 + \dots}}{\sqrt{A_0^2 + A_1^2 + A_2^2 + A_3^2 + A_4^2 + \dots}}$$

Il suffit au curseur de pointer chaque amplitude et de faire l'opération. On obtient, pour les premières raies :

1 kHz	2 kHz	3 kHz	4 kHz	5 kHz	6 kHz	7 kHz
13,4 V	404 mV	75 mV	55 mV	82 mV	7 mV	42 mV

Remarquons que de travailler avec les amplitudes et non les valeurs efficaces ne change pas le rapport.

→ **par la directive :**  
 .FOUR 1k V(RL)

qui donne le résultat de ce calcul dans le fichier de sortie.

On lit ainsi que le fondamental a une amplitude de 13,45 V, et que les harmoniques suivantes ont, respectivement, 0,414 V, 80 mV, 58 mV, 82 mV...

Valeurs très proches données par la FFT.

```
DC COMPONENT = 8.3097E-02
```

HARMONIC NO	FREQUENCY (HZ)	FOURIER COMPONENT	NORMALIZED COMPONENT	PHASE (DEG)	NORMALIZED PHASE (DEG)
1	1.0000E+03	1.3455E+01	1.0000E+00	-1.7964E+02	0.0000E+00
2	2.0000E+03	4.1468E-01	3.0819E-02	9.4949E+01	4.5422E+02
3	3.0000E+03	8.0507E-02	5.9834E-03	1.7222E+02	7.1113E+02
4	4.0000E+03	5.8103E-02	4.3182E-03	-9.3292E+01	6.2525E+02
5	5.0000E+03	8.2145E-02	6.1051E-03	1.4638E+01	9.1282E+02
6	6.0000E+03	5.6464E-03	4.1965E-04	1.2352E+02	1.2013E+03
7	7.0000E+03	4.1645E-02	3.0951E-03	2.6572E+01	1.2840E+03
8	8.0000E+03	9.4561E-03	7.0279E-04	-9.3455E+01	1.3436E+03
9	9.0000E+03	3.7818E-02	2.8106E-03	2.8192E+01	1.6449E+03

TOTAL HARMONIC DISTORTION = 3.2553E+00 PERCENT

Le résultat est : THD = 3,25 %

### Conclusion

Ce montage est un modeste amplificateur de puissance audio. Ses performances peuvent être améliorées par un choix de transistors MOS de puissance plus récents. Néanmoins cette analyse sert de base pour la compréhension du schéma.

**Annexe : netlist prête à simuler.**

permet la simulation polarisation (.OP) et temporelle (.TRAN) de tout le circuit.

Pour l'analyse harmonique (.AC) de Q1, il faut commenter la ligne R2, et décommenter les 3 lignes d'après.

Pour la détermination du taux de distorsion harmonique, valider la ligne correspondante.

```
Ampli audio à Push Pull CMOS
* fichier ampli_MOS.cir

.lib eval.lib ; contient les 3 transistors

VDD E 0 DC 40 ; alim puissance unique
Vin IN 0 sin(0 50m 1k) AC 1 ; 50 mV ou 0,6 V
C1 IN 1 220u
R1 1 2 1k
R6 2 0 1.8k
R3 5 0 100
C2 5 0 1000u
R2 2 6 27k
*Vpola 30 0 DC=21.968 ; 3 lignes
*R2 30 2 27k ; pour l'analyse
*.AC DEC 100 1 1Meg ; AC de Q1
R5 E 3 2.4k
R4 3 4 470
Q1 4 2 5 Q2N2222 ; C B E
R8 3 G1 330
R7 4 G2 330
M1 E G1 6 6 IRF150 ; NMOS D G S Sub
M2 0 G2 6 6 IRF9140 ; PMOS D G S Sub
C3 6 OUT 2200u
RL OUT 0 8
.OP
.TRAN 1u 20m 0m 1u ; 20m pour FFT
*.FOUR 1k V(RL) ; pour THD
.probe
.end
```

articles 1 à 43 : sur le livre

**Tableau récapitulatif des articles PDF disponibles sur ce site**

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	<a href="#">présentation</a>	<a href="#">document PDF</a>
44	Exemples basiques et des exercices...	<a href="#">présentation</a>	<a href="#">document PDF</a>
45	Un exemple de circuit passif	<a href="#">présentation</a>	<a href="#">document PDF</a>
46	Un oscillateur Colpitts	<a href="#">présentation</a>	<a href="#">document PDF</a>
47	Compensation en fréquence des amplificateurs opérationnels	<a href="#">présentation</a>	<a href="#">document PDF</a>
48	Un amplificateur à transistors bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
49	Une bascule D Flip Flop CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
50	Une porte XOR à transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
51	Un VCO à 12 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
52	Une PLL à moins de 20 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
53	Un oscillateur à résistance négative	<a href="#">présentation</a>	<a href="#">document PDF</a>
54	Une charge électronique	<a href="#">présentation</a>	<a href="#">document PDF</a>
55	Un amplificateur en classe C	<a href="#">présentation</a>	<a href="#">document PDF</a>
56	Le monostable 74 123	<a href="#">présentation</a>	<a href="#">document PDF</a>
57	Un amplificateur en classe D	<a href="#">présentation</a>	<a href="#">document PDF</a>
58	Le transformateur en linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
59	La loi d'ohm thermique	<a href="#">présentation</a>	<a href="#">document PDF</a>
60	Le transformateur en non linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
61	Robustesse d'un oscillateur en anneau	<a href="#">présentation</a>	<a href="#">document PDF</a>
62	Une alimentation stabilisée	<a href="#">présentation</a>	<a href="#">document PDF</a>
63	Modélisation d'un haut-parleur	<a href="#">présentation</a>	<a href="#">document PDF</a>
64	Un synthétiseur de fréquence	<a href="#">présentation</a>	<a href="#">document PDF</a>
65	Un ampli audio de Sparkfun	<a href="#">présentation</a>	<a href="#">document PDF</a>
66	Simulation logique et analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
67	Un oscillateur à relaxation	<a href="#">présentation</a>	<a href="#">document PDF</a>
68	Lecteur de TAG RFID 125 kHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
69	Diagramme de l'œil avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
70	Un amplificateur hautes fréquences	<a href="#">présentation</a>	<a href="#">document PDF</a>
71	Une bizarrerie enfin expliquée...	<a href="#">présentation</a>	<a href="#">document PDF</a>
72	Comprendre le paramétrage de la FFT	<a href="#">présentation</a>	<a href="#">document PDF</a>
73	La relation de Bennett	<a href="#">présentation</a>	<a href="#">document PDF</a>
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	<a href="#">présentation</a>	<a href="#">document PDF</a>
75	Une horloge biphasé sans recouvrement	<a href="#">présentation</a>	<a href="#">document PDF</a>
76	Quelques simulations sur la diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
77	Un ampli classe A, avec transformateur de sortie	<a href="#">présentation</a>	<a href="#">document PDF</a>
78	Des stimuli pour PSpice	<a href="#">présentation</a>	<a href="#">document PDF</a>
79	Simuler le TL431 : zener ajustable	<a href="#">présentation</a>	<a href="#">document PDF</a>
80	Un ADC flash	<a href="#">présentation</a>	<a href="#">document PDF</a>
81	Une chaîne d'acquisition : S&H, ADC, DAC	<a href="#">présentation</a>	<a href="#">document PDF</a>
82	Un amplificateur 50 MHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
83	Un dérivateur non inverseur	<a href="#">présentation</a>	<a href="#">document PDF</a>
84	Un amplificateur bipolaire avec push pull CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
85	Rôle des répéteurs logiques dans un circuit intégré	<a href="#">présentation</a>	<a href="#">document PDF</a>
86	Un driver logique CMOS pour charge 50 ohms	<a href="#">présentation</a>	<a href="#">document PDF</a>
87	Des triggers de Schmitt et des applications	<a href="#">présentation</a>	<a href="#">document PDF</a>
88	Un filtre gaussien analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
89	Un générateur de bruit rose	<a href="#">présentation</a>	<a href="#">document PDF</a>

90	Un anémomètre à fil chaud : simulation comportementale	<a href="#">présentation</a>	<a href="#">document PDF</a>
91	Un oscillateur à pont de Wien stabilisé par CTN	<a href="#">présentation</a>	<a href="#">document PDF</a>
92	L'emballement thermique d'une diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
93	Les puissances dans un amplificateur	<a href="#">présentation</a>	<a href="#">document PDF</a>
94	Asservissement de puissance dans une résistance	<a href="#">présentation</a>	<a href="#">document PDF</a>
95	Asservissement de la puissance émise par une antenne radio	<a href="#">présentation</a>	<a href="#">document PDF</a>
96	Un driver de LED de puissance	<a href="#">présentation</a>	<a href="#">document PDF</a>
97	Exploiter Pspice pour simuler des filtres numériques	<a href="#">présentation</a>	<a href="#">document PDF</a>
98	Un filtre en cosinus surélevé avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
99	Effet de la température sur un amplificateur en classe A	<a href="#">présentation</a>	<a href="#">document PDF</a>
100	Un amplificateur à transistors JFET et bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
<b>Supplément, hors article :</b>			
<b>mon cours « Electronique pour les communications numériques », <a href="#">polycopié couleur 201 pages en pdf</a></b>			

[retour à l'écran d'accueil de ce site](#)